Security Engineering: νActionGUI Tutorial

Contents

1 Introduction

In this tutorial, we introduce νActionGUI. It implements a specific instance of the model-driven security and privacy methodology taught in the lectures. In particular, the design modeling language used is a subset of the ComponentUML model. It combines the design modeling language metamodel with the SecureUML and PrivateUML metamodels, while exposing a textual concrete notation. It also allows generation of data-centric web applications that enforce fine-grained access control policies, as well as purpose limitation and data-subject consent privacy policies.

νActionGUI provides developers with both modeling primitives and model transformations to specify and automatically generate the resulting web application. Modeling primitives allow developers to specify the state space of the application (via the data model), fine-grained access control policies (via the security model), and privacy policies (via the privacy model). Model transformations map the data model to a set of classes with a database schema, the security model to role-based access control policies and programmatic authorization checks, and privacy model to consent collection, actual purpose tracking, and purpose-based access control mechanisms.

In this tutorial, we start with a simple example to introduce the workflow of νActionGUI. As a main running example for this tutorial, we consider the Thoughts application: a simple web application that allows *Persons* to create *Thoughts*, share them with others, and vote for their favorite thoughts.

2 Installation

To use νActionGUI, download the `NuActionGUI` file and extract it. Inside you will find a `docker-compose.yml` file containing a Linux instance with all the standard Python libraries installed.

Alternatively, you can install νActionGUI directly on your machine. This tutorial, however, will only assume that Docker[1] is installed on your machine and that you are running νActionGUI inside Docker. For those who decide to install the tool locally, you can skip to Section 3. With Docker installed, you can enter the downloaded `NuActionGUI` directory and type:

```
1 $ docker compose up --build
```

[1] `https://www.docker.com/get-started`

This will build the images and run a docker container. On subsequent usage of the container, you can omit `-build` flag when starting.

To check that the containers are running type `docker ps` command and you should see the container named `nag-app`.

3 A minimal Thoughts application

Inside the container, the root folder is linked to the local folder in your host machine. Any change in this root folder will be visible on your host machine (and vice-versa). Therefore, you can edit files on your host machine using your favorite code editor and use the container for compiling and running the project.

Navigate to the `models` folder and create a new project there:

```
$ cd /models
$ mkdir phil
```

If you want to name the project differently, change `phil` above to the desired name. You may need to adjust the permissions of the project files in the container (e.g., `chmod -R a+w phil`) in order to edit them on the host machine.

We will introduce the rest of the development workflow (e.g., generation, compilation, deployment, etc.) through examples. The important parts of the workflow are summarized in Appendix A as a reference.

We now provide instructions for writing, compiling, and running a minimal "Hello World!" application. Given the newly created νActionGUI project with name `phil`, we need to create three models:

1. a data model (Section 3.1)
2. a security model (Section 3.2), and
3. a privacy model (Section 3.3).

3.1 Data Model

1. Create the `/models/phil/project.dtm` file.
2. Copy the code below and save it. Click here for easy-to-copy version.

```
entity Person {
  String name
  Set(Thought) created in Person_thought
  Set(Thought) voted in Vote
  @entry add_thought(String content)
  @entry delete_thought(Integer thoughtId)
  @entry vote_thought(Integer thoughtId)
}

entity Thought {
  String content
  Person createdBy in Person_thought
  Set(Person) votedBy in Vote
  getAllThoughts()
}
```

This datamodel is similar to the ComponentUML diagram in Figure 1



Figure 1: Thought ComponentUML diagram

The data model defines the state space of the application. One can think of it as defining the classes and attributes in an object-oriented programming language like Java or C++.

The νActionGUI language for modeling application domains is a textual language, which provides a subset of the ComponentUML metamodel (from the lectures), which is in turn a subset of the UML class diagram. In particular, the main modeling elements of the νActionGUI data model language are:

- **Entities**: They model the data in the application domain.
- **Attributes**: They model the properties of the entities.
- **Association-ends**: They model the associations between entities.
- **Methods**: They model possible methods of the system.

An entity is declared using the `entity` keyword. It may contain attributes and association ends. They are always declared with a name and a type (either `Boolean`, `Integer`, `String`, or a collection type).

An entity can also contains methods. There are two types of methods:

1. methods that represent potential entrypoints of the application (i.e., methods that can handle an HTTP request), and
2. methods that are invoked within the application.

The former one has a special keyword `@entry` as prefix.

Next we proceed to create a minimal security model for the data model.

3.2 Security Model

1. Create `/models/phil/project.stm` file.
2. Copy the code below and save the `project.stm` file.

```
USER Person

anonymous Role ANONYMOUS {
  Person {
    fullAccess
  }
  Thought {
    fullAccess
  }
}

default Role NORMALUSER {
  Person {
    fullAccess
  }
  Thought {
    fullAccess
  }
}
```

In this minimal application, we declare `Person` as the application user entity. Then we define two roles: `ANONYMOUS` and `NORMALUSER`. The `ANONYMOUS` role is intended for unauthenticated users, while the `NORMALUSER` role is the default role assigned when an unauthenticated user registers. Both roles are granted full access to both entities `Person` and `Thought`.

The action `fullAccess` is a composite action that consists of `create` and `delete` simple actions and `read` and `update` composite actions. The `read` and `update` composite actions allow reading and updating of all attributes and association ends, respectively.

A security model defines the application's authorization policy in terms of allowed actions on the resources provided by the data model. The νActionGUI language for modeling the application's security is a textual language that provides the following elements:

- **User entity**: The user entity represents the type of instances that interact with the application. In the security model, we assign at most one entity from the defined data model as the user class.
- **Roles**: Named in all-caps, they model a job or a function within the system. Each role groups all the permissions needed to fulfill the respective function. Users of the system can have one of the roles.
- **Permissions**: Each permission represents the authorization to execute an operation on some of the objects in the system. Permissions are grouped by the entities from the data model. They are of the form `action(subresource) [constraint]` where subresource can be the attribute or association end of the entity whose permission is grouped under and constraint is an Object Constraint Language (OCL) formula. If the subresource is omitted (together with the parentheses `( )`) the action is a composite action that applied to all the subresources of the resource. If the constraint is omitted (together with the brackets `[ ]`) it is assumed to be the `true` OCL expression.

The roles are declared with the keyword `Role` followed by the role's name and its permissions. At most one permission per entity from the underlying data model may be declared.

- The keyword **anonymous** before a role name declares that this is the role of unauthenticated users.
- The keyword **default** before a role name declares that this is role a user gets once they register to the application.
- The keyword **extends** followed by an existing role after a role name specifies role inheritance.

Finally, we define the privacy model for this minimal application.

3.3 Privacy Model

1. Create `/models/phil/project.ptm` file.
2. Copy the code below and save to this `project.ptm` file.

```
1 Personal data {}
2 Purposes {}
3 Actual purposes {}
4 Declared purposes {}
```

A privacy model identifies application's personal data and purposes. It declares purposes for which personal data will be used and under what conditions. Finally, it associates data model's methods with purposes.

νActionGUI performs purpose-based access control based on the privacy model. Any action performed on a personal data within some method is checked: the method's (actual) purpose is compared to the data's (declared) purpose. Furthermore, a data owner's consent to any of the declared purposes is taken into account when performing the check.

The privacy model contains the following elements:

- **Personal data**: Subresources of the data model that are considered to store personal user data. They are declared using the `Resource.subresource` syntax.
- **Purposes**: Used in the application named in all-caps. A purpose declared on its own is a simple purpose. A purpose followed by the **includes** keyword and a list of purposes is a complex purpose.
- **Actual purposes**: Declare a mapping from methods in the data model to purposes using the syntax `Resource.method purpose`.
- **Declared purposes**: Specify the purposes for which each personal data is used and under what condition. It uses the syntax `Resource.subresource purpose [constraint] <description>` where constraint is an OCL formula and description is a string describing it. The constraint and the description can both be omitted, in which case they are `true` and empty string, respectively.

In this minimal application we do not have any purposes.

3.4 First time: Generate and Build

Generating the project First, open a shell to the `nag-app` container:

```
$ docker exec -it nag-app bash
```

Once you have those models defined, navigate to the `/src` folder and create a virtual environment, activate it, and install necessary Python packages:

```
$ python3 -m venv .venv
$ . .venv/bin/activate
(.venv) $ pip3 install -r ../requirements.txt
```

To generate the project for the first time, execute:

```
(.venv) $python3 generate.py -p phil
```

Replace `phil` with the desired project name from the projects folder `/models`. This will create a new folder `phil` in the output folder `/output`. Optionally, one can declare the desired output folder by adding the flag: `-o` and the name of the folder, then the generated application will be located in `/<foldername>`.

Running the project To execute the compiled web application, go to the `/output/phil` folder:

```
(.venv) $ flask --app app.py run --host=0.0.0.0
```

The app should be reachable on your local machine via `localhost:5000`. In this version, there is no implemented functionality as the generation only creates a backbone Flask application. Figure 2 displays what the generated interface should look like: it signals you to implement the generated methods stubs.

3.5 Implementing the functionality

We now implement the business logic of the method stubs and the corresponding HTML pages.

3.5.1 Implementing the HTML pages

In the folder `/output/phil`, replace the file `templates/main.html` with the following code. This snippet of code implements the interfaces required



Figure 2: Snapshot of the application index page

for the Thoughts application, including the table element that contains all thoughts, in which, each thought has a creator, a content and a number of vote fields. Furthermore, each thought has a button to vote and another button to delete it. The interface is very similar to the ones introduced in the previous Flask tutorials.

3.5.2 Implementing the business logic

In the folder where we generated the web application, we first locate the file `project.py` containing the generated method stubs. Notice that each method in `project.py` returns a string `TODO: Implement the method stub` (e.g., line 3 below).

```
def add_thought(request):
    # TODO: implement the method stub
    return "TODO: Implement the method stub"
```

To begin with, import the following lines at the beginning of the file:

```
from dtm import Thought, vote
from app import db
from auxiliary import getAllThoughts
```

Implementing `main()` As the main index page displays all thoughts, we extend the implementation of the `main()` method as follows:

```
def main(request):
    thoughts = getAllThoughts()
    return render_template('main.html', table=thoughts, user=
        current_user)
```

The snippet of code can be copied from here.

As this function calls `getAllThoughts()` from generated `auxiliary.py`, which in turn calls functions from the `project_aux.py`. Open `auxiliary.py` and copy the below code:

```
from dtm import Thought, vote

def getAllThoughts():
    return Thought.query.all()
```

Implementing `add_thought()` To add a thought, we first obtain the `content` from the request (line 2). We check whether the content is valid (line 3), then create the Thought object (line 4), update the content of the newly created thought object with the valid content (line 5), update the creator of this thought to the current user (line 6), add this thought, and commit to the database (line 7-8).

```
def add_thought(request):
    content = request.form["thought"]
    if content != "":
        t = Thought()
        t.content = content
        t.createdBy = current_user
```

```
7           db.session.add(t)
8           db.session.commit()
9           return redirect(url_for('main'))
10      else:
11          thoughts = Thought.query.all()
12          return render_template('main.html', table=thoughts,
                invalid_input=True)
```

The snippet of code can be copied from here.

Implementing `delete_thought()` To delete a thought, we first obtain the id from the context (line 2). Then we obtain the thought from the database (line 4). If the thought is valid (line 5), we delete it (line 6).

```
1  def delete_thought(request):
2      id = request.args["id"]
3      i = int(id)
4      t = Thought.query.get(i)
5      if t != None:
6          t.__delete__(db)
7          db.session.commit()
8      return redirect(url_for('main'))
```

The snippet of code can be copied from here.

Implementing `vote_thought()` Similarly, to vote for a thought, we obtain the voted thought id from the context (line 2) and get this thought from the database (line 4). If the thought and the current user are valid (line 6), we proceed to create the vote by either adding the user object to the votedBy list of the thought (line 7) or adding the thought object to the voted list of the current user (line 8).

```
1  def vote_thought(request):
2      id = request.args["id"]
3      i = int(id)
4      t = Thought.query.get(i)
5      c = current_user
6      if t != None and c != None:
7          t.votedBy.append(c)
8          # c.voted.append(t)
9          db.session.commit()
10     return redirect(url_for('main'))
```

The snippet of code can be copied from here.

3.6 Recompiling the project

Now that we have implemented the Thoughts application, we run the application again by navigating to the `output/phil/` folder and typing:

```
(.venv) $ flask --app app.py run --host=0.0.0.0
```

Figure 3 displays what generated interface should now look like.



Figure 3: Snapshot of the application index page

There is currently no security or privacy enacted. You can go ahead and test the application by registering as a new user, logging in, creating new thoughts, voting, and deleting them. In the upcoming sections, we will implement some security and privacy policies for this application.[2]

[2]There is currently a redirecting problem when a user logs in. If you encounter this problem, refresh the page. We are working on the patch for this.

4 Thoughts Application: Security

In this section, we focus on enforcing the following security policies:

(SP1) a visitor (i.e., user with `ANONYMOUS` role) cannot perform any action except for reading the content of thoughts.

(SP2) a normal user (i.e., user with `NORMALUSER` role) can do whatever a visitor can do and see other users' names. They can change only their name and thoughts' content;

(SP3) a normal user can vote at most three times for the same thought; and

(SP4) a normal user is not allowed to vote for their own thoughts.

(SP5) a normal user can create and read thoughts, but delete only their own thoughts. They may initialize the thought's owner to themselves and update the content of their thoughts, or of thoughts with uninitialized owner.

4.1 Adjust the security model

We now update the security model in `project.stm` to reflect the aforementioned requirements:

4.1.1 Anonymous role

First, we update the `ANONYMOUS` role to reflect (SP1). We add a read permission on the `content` attribute of a `Thought` (i.e., `read(content)`).

```
anonymous Role ANONYMOUS {
  Person {}
  Thought {
    read(content)
  }
}
```

4.1.2 Normal role

- We update the `NORMALUSER` role to reflect (SP2). Specifically, it extends the `ANONYMOUS` role, hence it can also read the thoughts' contents. It can additionally read the names of the users (due to the `read(name)` permission in the `Person`). Finally, a `NORMALUSER` can update the `name` attribute of the `Person` entity if the updated entity (`self`) is exactly the user executing the action (`caller`). When updating the `content` attribute of a `Thought` entity `self` the thought's `createdBy` attribute should be equal to the user executing the action.

```
default Role NORMALUSER extends ANONYMOUS {
  Person {
    read(name)
    update(name) [self = caller]
  }
  Thought {
    update(content) [self.createdBy = caller]
  }
}
```

- We update the `NORMALUSER` role to reflect (SP3). Namely, when a user adds their vote to the association we count how many votes from the user are already in the association. Note that we write two "symmetrical" constraints on both `voted` and `votedBy` association ends.

```
default Role NORMALUSER extends ANONYMOUS {
  Person {
    read(name)
    update(name) [self = caller]
    add(voted) [value.votedBy->select(v|v = caller)->size() < 3]
  }
  Thought {
    update(content) [self.createdBy = caller]
    add(votedBy) [self.votedBy->select(v|v = caller)->size() < 3]
  }
}
```

- We update the `NORMALUSER` role to reflect (SP4). The two constraints from the previous point are each extended with a conjunct that checks that the thought is not created by the voting user.

```
default Role NORMALUSER extends ANONYMOUS {
  Person {
    read(name)
    update(name) [self = caller]
    add(voted) [value.votedBy->select(v|v = caller)->size() < 3
        and value.createdBy <> caller]
  }
  Thought {
    update(content) [self.createdBy = caller]
    add(votedBy) [self.votedBy->select(v|v = caller)->size() < 3
        and self.createdBy <> caller]
  }
}
```

- Finally, we update the security model to account for (SP5). We add create, delete and read permissions on `Thought` entity, an update permission on `createdBy` attribute, and extend the constraint in the `content` attribute's update permission. The final security model is as follows:

```
USER Person

anonymous Role ANONYMOUS {
  Person {}
  Thought {
    read(content)
  }
}

default Role NORMALUSER  extends ANONYMOUS {
  Person {
    read(name)
    update(name) [self = caller]
    add(voted) [value.votedBy->select(v|v = caller)->size()<3
        and self <> caller]
  }
  Thought {
    create
    delete [self.createdBy = caller]
    read
    update(createdBy) [self.createdBy = null and value = caller]
    update(content) [self.createdBy = null
        or self.createdBy = caller]
    add(votedBy) [self.votedBy->select(v|v=caller)->size() < 3
        and self.createdBy <> caller]
  }
}
```

The snippet of this code can be copied from here.

A couple of remarks are in order. First, we had to specify more security policies here (e.g., the policy (SP5)) compared to the Thoughts application from the tutorial on authorization in Flask. This is because, unlike the implementation in the previous tutorial, νActionGUI employs deny-by-default approach, preventing the proper functionality of the application without these additional policies. This is not a drawback of νActionGUI, but rather the policies in the previous tutorial are inherently incomplete and the final application is insecure. Second, notice that the `NORMALUSER` can read thoughts'

contents via two separate permissions: one inherited from the `ANONYMOUS` role and the other from the read permission on the entire `Thought` entity. One could decide to refactor this redundancy and reflect it back in the natural language security policies. Finally, note that upon the creation of a thought, the order of initialization of the attributes `createdBy` and `content` does not matter due to the `self.createdBy = null` OCL term. In other cases, the order may be important: for example, if the OCL term were not present in the `update(content)` permission.

4.2 Recompiling the project

Since we have updated the security model, we need to regenerate parts of the system. To do so, navigate to the `src` folder and call:

```
(.venv) $ python3 generate.py -p phil -o output -re
```

This will regenerate part of the system, those that are related to the security and privacy model. The implemented `project.py` and `project_aux.py`, as well as the HTML template `main.html` will not be changed.

Before running the application after the regeneration, remove the current database in `output/phil/instance/app.db`, as it may no longer be valid.

Now that we have enforced some security policies to the Thoughts application, we rebuild the the application in the `output/phil/` folder using:

```
(.venv) $ flask --app app.py run --host=0.0.0.0
```

In this new version, we have implemented its functionality and the aforementioned security. Since we did not change any related to the business logic and the HTML interface, the webpage looks the same.

Nevertheless, when opening the webpage as a visitor, we observe that one can only read the content of the thought, but not the value of the other fields (see Figure 5). Moreover, the create, vote and delete functions have been revoked from the visitors (see Figure 4).

As a normal user, upon logging in, the user should be able to read everything related to thoughts (see Figure 6). In addition, the user should be able to create a new thought, vote for thoughts (but one cannot vote more than three times for a thought or for their own thought), and delete (its own) thoughts.[3]

[3]There is currently a redirect issue when a user logs in. If you encounter this problem, refresh the page. We are working on a patch for it.



Figure 4: Snapshot of the application when a visitor try to delete a thought



Figure 5: Snapshot of the application when a visitor tries to visit the website.



Figure 6: Snapshot of the application when a user tries to visit the website.

5 Thoughts Application: Privacy

In νActionGUI, we focus on the following requirements:

- Purpose-limitation (Art. 5 §1 (b) GDPR): "Personal data shall be collected for specified, explicit, and legitimate purposes and not further processed in a manner that is incompatible with those purposes."
- Data-subject consent (Art. 7 §1 GDPR): "Where processing is based on consent, the controller shall be able to demonstrate that the data subject has consented to the processing of his or her personal data."

We now define the privacy model to support the enforcement of the privacy requirements.

5.1 Adjust the privacy model

5.1.1 Personal data

Let us consider the following subset of the data model as personal data: `Person`'s attributes, such as name and role; association-ends, such as created and voted, and `Thought`'s votedBy and createdBy are personal data. Let's replace the first part of the privacy model `project.ptm` with the following:

```
Personal data {
    Person.name,
    Person.role,
    Person.created,
    Person.voted,
    Thought.createdBy
    Thought.votedBy
}
```

Since we consider all subresources of the `Person` entity as personal data, one can also use the following shorthand as syntactic sugar:

```
Personal data {
    Person,
    Thought.createdBy,
    Thought.votedBy
}
```

5.1.2 Actual purposes

Let us now associate some application's functionalities to purposes, namely: `ADD_NEW_THOUGHT`, `VOTE` (for thoughts) and `DISPLAY` (thoughts). Below we modify the second part of the privacy model to define the purposes, and the third part to associate the purposes with the methods from the data model:

```
Purposes {
    DISPLAY,
    VOTE,
    ADD_NEW_THOUGHT
}

Actual purposes {
    Person.add_thought ADD_NEW_THOUGHT,
    Person.vote_thought VOTE,
    main DISPLAY
}
```

5.1.3 Declared purposes

We declare the following privacy policies:

(PP1) We use your `voted` attribute for the purpose of voting for thoughts.

(PP2) We use your `created` attribute for the purpose of adding new thoughts.

(PP3) We use your `name` and your thoughts' `votedBy` and `createdBy` attributes for the purpose of displaying thoughts.

The following part of the privacy model captures the aforementioned policies:

```
Declared purposes {
    Person.voted VOTE,
    Thought.votedBy VOTE,
    Person.created ADD_NEW_THOUGHT,
    Thought.createdBy ADD_NEW_THOUGHT,
    Person.name DISPLAY,
    Thought.votedBy DISPLAY,
    Thought.createdBy DISPLAY
}
```

The complete privacy model can be copied from here.

5.2 Recompiling the project

Once the privacy model is defined, let us rebuild the web application by navigating to `output/phil` and calling:

```
(.venv) $ python3 generate.py -p phil -o output -re
```

Before running the application after the regeneration, remove the current database in `output/phil/instance/app.db`, as it may no longer be valid. Since we updated the privacy model, we need to regenerate this part of the system. To do so, navigate to the `src` folder and call:

```
(.venv) $ flask --app app.py run --host=0.0.0.0
```

In this new version, we have implemented the functionality, along with the aforementioned security in Section 4, and a privacy policy in this section that complies with the privacy requriements of *purpose limitation* and *data consent*. The webpage looks the same since we did not change the implementation inside `project.py` and `main.html`.

However, when one attempts to create a thought as a `NORMALUSER` user, the system will prevent the user from doing so. This is because the user has not opted in for to give consent for using his information for the purpose of create new thoughts (i.e., `ADD_NEW_THOUGHT`). Figure 7 shows the webpage's redirection.



Figure 7: Snapshot of the application when a user tries to create a new thought without giving proper consent.

One can access the `policy` endpoint (i.e., `localhost:5000/policy`) and opt in for giving consent to use personal data to create new thoughts. Figure 8 shows the interface of the privacy policy. After this, the user will be able to create new thoughts.



Figure 8: Snapshot of the application privacy policy

A Development Workflow: Checklist

To start for the first time make sure you have Docker installed. You can download the latest version of Docker from here.

In order to use νActionGUI, download the `ActionGUI` file from the course's Moodle page and extract it. The `docker-compose.yml` file contains a Linux instance with python standard libraries installed.

To start for the first time, navigate to the directory that contains the `docker-compose.yml` file, and type:

```
$ docker compose up --build
```

This will build the images and run the container (named `nag-app`) for the first time. Once the image has been built, you can omit `-build` flag when starting.

Then, to open a shell to the `nag-app` container:

```
$ docker exec -it nag-app bash
```

Here we assume you have installed νActionGUI using Docker, started the containers, and have a shell in the `nag-app` container.

Since the folder `models/` in the container and in the local are synced, you can edit your models locally using your favorite code editor and see these changes reflected on the container.

For a new project, create a sub-directory in the folder `models/`. In this sub-directory, you can define your models (data, security, and privacy models).

Generating the project Once you have the models defined, go to the `src/` folder and build a virtual environment in python and install necessary packages inside this environment:

```
$ python3 -m venv .venv
$ . .venv/bin/activate
(.venv) $ pip3 install -r ../requirements.txt
```

Then, to generate the project for the first time, in the virtual environment:

```
(.venv) $ python3 generate.py -p <project_name> -o <
   output_folder>
```

Replace `<project_name>` with the desired project name from the `models/` folder. This will create a new folder `<project_name>` in the output folder `<output_folder>/`. The `-o` option is optional, by default, the output folder will be `output`. You may need to adjust the permissions of the bash script in the container (e.g., `chmod -R +x run.sh`) in order to execute it on the host machine.

Regenerate part of the project For now, we assume the data model is fixed and will not be changed during the tutorial. If the security model and/or the privacy model in the `models/<project_name>` changes, to regenerate only the related artifacts, go to the `src/` folder:

```
(.venv) $ python3 generate.py -p <project_name> -re
```

Running the project To execute the compiled web application, go to the `/<output_folder>/<project_name>` folder:

```
(.venv) $ flask --app app.py run --host=0.0.0.0
```

The app should be reachable from your local machine at `localhost:5000/`.